Creep Curve

Univ.-Prof. Dr.-Ing. habil. J. BETTEN
RWTH University Aachen
Mathematical Models in Materials Science and Continuum Mechanics
Augustinerbach 4-20
D-52056 A a c h e n , Germany

betten@mmw.rwth-aachen.de

Abstract

Using MAPLE V, Release 10, a *creep curve* has been formulated, which represents the mechanical behavior of metals, polymers, and ceramics, for instance.

Keywords: exponential description; primary, secondary, and tertiary creep stage; creep rate and acceleration

Introduction

Creep tests are carried out on specimens loaded, e.g., in tension or compression, usually at constant load, inside a furnace which is maintained at a constant temperature. The extension of the specimen

is measured as a function of time. A typical *creep curve* for metals, polymers, and ceramics exsists of three parts and is represented in this Maple worksheet. The *primary or transient creep* is characterized by a monotonic decrease in the rate of creep. Creep deformations of the *secondary* stage are large and of a similar character to "pure" plastic deformations. The *tertiary creep phase* is accompanied by the formation of microscopic cracks on the grain boundaries, so that damage-accumulation occurs.

In the past three decades there has been considerable progress and significant advances made in the development of fundamental concepts of creep and damage mechanics an their application to solve practical engineering problems (BETTEN, J.: Creep Mechanics, 2nd Edition, Springer - Verlag, Berlin / Heidelberg / New York 2005).

Exponential Description

```
> restart:
  epsilon[creep](t):=A[11]*(1-exp(-A[12]*sqrt(t)))+ A[21]*t+A
  [31]*(exp(A[32]*t^n)-1);
```

$$\varepsilon_{creep}(t) := A_{11}\left(1 - e^{-A_{12}\sqrt{t}}\right) + A_{21}\,t + A_{31}\left(e^{A_{32}\,t^{n}} - 1\right) \quad \textbf{(2.1)}$$

```
> Digits:=5:
```

```
> epsilon[c](t):= subs({A[11]=0.4,A[12]=5,A[31]=0.02,A[32]=3,n=
  10},(2.1));
```

$$\varepsilon_c(t) := 0.38 - 0.4\,e^{-5\sqrt{t}} + A_{21}\,t + 0.02\,e^{3\,t^{10}} \quad \textbf{(2.2)}$$

```
> epsilon[c](0):=evalf(subs(t=0,(2.2)));
```

$$\varepsilon_c(0) := 0. \quad \textbf{(2.3)}$$

```
> epsilon[c](1):=evalf(subs(t=1,(2.2)));
```

$$\varepsilon_c(1) := 0.77902 + A_{21} \quad \textbf{(2.4)}$$

```
> A[21]:=solve(epsilon[c](1)=1,A[21]);
```

$$A_{21} := 0.22098 \quad \textbf{(2.5)}$$

```
> alias(H=Heaviside,th=thickness):
> plot1:=plot(epsilon[c](t),t=0..1,color=black,th=2):
> plot2:=plot({epsilon[c](1),epsilon[c](1)*H(t-1)}, t=0..1.001,
  color=black, title="Creep Curve # exponential description"):
> plots[display]({plot1,plot2});
```



Time Derivative

```
> time_derivative(t):=diff(epsilon[c](t),t);
```

$$time\_derivative(t) := \frac{1.0000\,e^{-5\sqrt{t}}}{\sqrt{t}} + A_{21} + 0.60\,t^{9}\,e^{3\,t^{10}} \quad \textbf{(3.1)}$$

```
> time_derivative(0):=infinity;
```

$$time\_derivative(0) := \infty \quad \textbf{(3.2)}$$

```
> time_derivative(1):=evalf(subs({A[21]=0.22098,t=1},%%));
```

$$time\_derivative(1) := 12.280 \quad \textbf{(3.3)}$$

```
> plot3:=plot(time_derivative(t),t=0..1,0..2,color=black,th=2,
  title="Time Derivative of the Creep Curve"):
> plot4:=plot({2,2*H(t-1)},t=0..1.001,color=black):
> plots[display]({plot3,plot4});
```




```
> plots[display]({plot1,plot2,plot3,plot4});
```



Creep Parameters

The creep curve exsists of three parts:

```
> restart:
  parameters_of_the_primary_term:=A[11], A[12];
```

$$parameters\_of\_the\_primary\_term := A_{11}, A_{12} \quad \textbf{(4.1)}$$

```
> parameters_of_the_scondary_term:=A[21]=K*sigma^m;    # NORTON-
  BAILEY
```

$$parameters\_of\_the\_scondary\_term := A_{21} = K\,\sigma^{m} \quad \textbf{(4.2)}$$

```
> parameters_of_the_tertiary_term:=A[31],A[32],n;
```

$$parameters\_of\_the\_tertiary\_term := A_{31}, A_{32}, n \quad \textbf{(4.3)}$$

For the *primary creep* the sqrt(t)-law has been assumed [*BETTEN, J*.: Creep Mechanics, 2nd Edition, Springer-Verlag, Berlin / Heidelberg / New York, 2005].
The exponent n in the tertiary term regulates the tangent of the creep curve at the creep rupture time.

The creep parameters can be determined by suitable Experiments, some of which are discussed in the above mentioned book.

Creep Rate and Acceleration

```
> restart:
> epsilon[creep](t):=A[11]*(1-exp(-A[12]*sqrt(t)))+ A[21]*t+A[31]*(exp(A[32]*t^n)-1);
```

$$\varepsilon_{creep}(t) := A_{11}\left(1 - e^{-A_{12}\sqrt{t}}\right) + A_{21}\, t + A_{31}\left(e^{A_{32}\, t^n} - 1\right) \qquad \textbf{(4.1.1)}$$

```
> creep_rate(t):=diff(epsilon[creep](t),t);
```

$$creep\_rate(t) := \frac{1}{2}\,\frac{A_{11}\, A_{12}\, e^{-A_{12}\sqrt{t}}}{\sqrt{t}} + A_{21} + \frac{A_{31}\, A_{32}\, t^n\, n\, e^{A_{32}\, t^n}}{t} \qquad \textbf{(4.1.2)}$$

```
> creep_rate(0):=infinity;
```

$$creep\_rate(0) := \infty \qquad \textbf{(4.1.3)}$$

```
> creep_rate(1):=subs(t=1,(4.1.2));
```

$$creep\_rate(1) := \frac{1}{2}\, A_{11}\, A_{12}\, e^{-A_{12}} + A_{21} + A_{31}\, A_{32}\, n\, e^{A_{32}} \qquad \textbf{(4.1.4)}$$

```
> Digits:=5:
  creep_rate(1):=evalf(subs({A[11]=0.4,A[12]=5, A[21]=0.22098,
  A[31]=0.02,A[32]=3,n=10},(4.1.4)));
```

$$creep\_rate(1) := 12.280 \qquad \textbf{(4.1.5)}$$

```
> Creep_rate(t):=evalf(subs({A[11]=0.4,A[12]=5, A[21]=0.22098,
  A[31]=0.02,A[32]=3,n=10},creep_rate(t)));
```

$$Creep\_rate(t) := \frac{1.0000\, e^{-5.\sqrt{t}}}{\sqrt{t}} + 0.22098 + 0.60\, t^9\, e^{3.\, t^{10}} \qquad \textbf{(4.1.6)}$$

```
> acceleration(t):=diff(epsilon[creep](t),t$2);
```

$$acceleration(t) := -\frac{1}{4}\,\frac{A_{11}\, A_{12}\, e^{-A_{12}\sqrt{t}}}{t^{3/2}} - \frac{1}{4}\,\frac{A_{11}\, A_{12}^2\, e^{-A_{12}\sqrt{t}}}{t} + \frac{A_{31}\, A_{32}\, t^n\, n^2\, e^{A_{32}\, t^n}}{t^2} - \frac{A_{31}\, A_{32}\, t^n\, n\, e^{A_{32}\, t^n}}{t^2} + \frac{A_{31}\, A_{32}^2\, (t^n)^2\, n^2\, e^{A_{32}\, t^n}}{t^2} \qquad \textbf{(4.1.7)}$$

```
> Acceleration(t):=evalf(subs({A[11]=0.4,A[12]=5, A[31]=0.02,A
  [32]=3,n=10},(4.1.7)));
```

$$Acceleration(t) := -\frac{0.50000\, e^{-5.\sqrt{t}}}{t^{3/2}} - \frac{2.5000\, e^{-5.\sqrt{t}}}{t} + 5.40\, t^8\, e^{3.\, t^{10}} + 18.00\, t^{18}\, e^{3.\, t^{10}} \qquad \textbf{(4.1.8)}$$

```
> alias(H=Heaviside, th=thickness):
> plot1:=plot({Creep_rate(t),Acceleration(t)}, t=0..1,-10..10,
  color=black,th=2):
> plot2:=plot({10,-10,10*H(t-1),-10*H(t-1)}, t=0..1.001,color=
  black, title="Creep Rate  and  Acceleration"):
> plots[display]({plot1,plot2});
```